
Auto_Corner_Checker

*
*                                    ^
*                                   /_\
*               _____________________|                         ____/\RL___
*               |            |      _|_                      _|_  10K\/   |
*                ->        <-      /VCC\                    /VL \         |
*               q13`|____|'q12     \___/                    \___/         |
*                _ '|  | |`_         |                        |           |
*               |      |    |        |    R1 60u/1u @ 1k/sq  _|_          |
*               |vbq15 |____|       _|_   R2 60u/1u @ 1k/sq  ///          |
*               |_____      |vbq12  ///   R3 60u/1u @ 1k/sq               |
*               |     |     |________     R4 6u/1u  @ 1k/sq               |
*               |_    |    _|        |                                    |
*             q15 `|__|__|'q16       |______________________              |
*               <-'|     |`->  10X   |                      |      ^      |
*               |           |        |                      |     /_\     |
*               |_____/\r4__|    ___/|\____________________/|\_____|VCC   |
*                  | 6K \/      |    |                |     |      |      |
*                  |_    veq16  ->   |                 ->   |      |      |
*                 ///           q0`|_|                q11`|_|      |      |
*                               _ '|                   _ '|        |      |
*                ___           |                      |            |      |
*  _____________|INP|_________/|\_______________      |            |      |
*  |            |___|      ____|__________      |     |   vbq9     |      |
*  |                      |   vcq0        |     |     |_______     |      |
*  |             ___    <-                 ->   |     |   _|  |   _|      |
*  |    ____    |INN|_|'q1                 q2`|_|     |_|'q7  |_|'q9  10X |
*  |  _|_   |   |___| |`_                  _ '|         |`->    |`->     _|_
*  | /VAC\  |      |     |                |_______||__    |veq7   |_____|OUT|
*  | \___/  |      |     |                |       ||c1|  <-       |     |___|
*  |  _|_   |      |     |                |     ^     |_|'q8    <-        |
*  | /VIN\  |      |     |vcq1       vcq2 |    /_\    | |`_   _|'q10      |
*  | \___/  |      |     |_____           |    _|     |____|_| |`_   10X  |
*  |___|    |      |     |     |          |__|'q5     |           |       |
*     _|_   |      |     |_    |         _|  |`->    _| vbq10     |       |
*    /VCM\  |      |   q3  `|__|_______|'q4     |__|'q6           |       |
*    \___/  |      |     <-'|          |`-> vbq6|  |`->           |       |
*      |    |      |     |                |     |     |___________|       |
*      |    |      |     |__/\r1____/\r2__|     |/\r3_|                   |
*     _|_   |      |      60K \/ |60K \/       60K \/ |                   |
*     ///   |      |    veq3     |_    veq4          _|_                  |
*           |      |            ///                  ///                  |
*           |      |                                                      |
*           |_/\r5_|_____________/\r6_____________________________________|
*            10K\/             10K \/

*==============================NETLIST=============================================
*==================================================================================
.OPTIONS  GMIN=1e-15    METHOD=gear   ABSTOL=1e-15  TEMP=27   ITL1=600    
*======== ====== ====== ====== ====== ====== ====== 
*V_DC_NUM NODE_P NODE_N        DC     VALUE
*======== ====== ====== ====== ====== ====== ====== 
VCC       VCC    0             DC     5
VCM       INP    0             DC     1.5
VIN       VIN    INP           DC     1u     
VAC       VAC    VIN           AC     1m    DC     0
VL        VL     0             DC     1
*======== ====== ====== ====== ====== ====== ====== ====== ======
*Q_NUMB   NCOL   NBASE  NEMIT  NSUB   MODEL  AREA
*======== ====== ====== ====== ====== ====== ====== ====== ======
q13       vbq15  vbq12  Vcc           pnpl   1
q12       vbq12  vbq12  Vcc           pnpl   1 
q15       vbq15  vbq15  0             npnv   1
q16       vbq12  vbq15  veq16         npnv   10
q0        vcq0   vbq12  Vcc           pnpl   1
q11       vbq9   vbq12  Vcc           pnpl   3 
q1        vcq1   INN    vcq0          pnpl   1
q2        vcq2   INP    vcq0          pnpl   1 
q3        vcq1   vcq1   veq3          npnv   1
q4        vcq2   vcq1   veq4          npnv   1
q5        Vcc    vcq2   vbq6          npnv   1
q6        vbq10  vbq6   0             npnv   3
q7        vbq9   vbq9   veq7          npnv   1
q9        Vcc    vbq9   OUT           npnv   10
q8        vbq10  vbq10  veq7          pnpl   1
q10       0      vbq10  OUT           pnpl   10 
*======== ====== ====== ====== ====== ====== ====== ====== ======
*R_NUMB   NODE1  NODE2  RVALUE        MODEL  L=VAL  W=VAL 
*======== ====== ====== ====== ====== ====== ====== ====== ======
r1        veq3   0                    RMODEL L=60u  W=1u
r2        veq4   0                    RMODEL L=60u  W=1u
r3        vbq6   0                    RMODEL L=60u  W=1u
r4        veq16  0                    RMODEL L=6u   W=1u
c1        vcq1   vbq10  5p
r5        VAC    INN    10k         
r6        OUT    INN    10k    
rl        OUT    VL     10k

*=======================SIMULATION=====================================================
* 2 SupplyVoltages X 3 Temperatures X 2 PNPL_beta X 2 ResOhmperSq = 24 Datasheet tests
*======================================================================================

.control
set outfile =    "OpAmpCornerData.txt"
echo                   "OpampCornerTester"    >      $outfile

foreach processval    1 2 3                 
echo                   "=============================PROCESS_CHANGE=================================" >> $outfile

if ( $processval =      2                                        )
alter @rmodel[rsh] = 1100
end                     ***<-- if $processval

if ( $processval =      3                                        )
alter @pnpl[bf] =    120
end                     ***<-- if $processval
let npnbf  =            @npnv[bf]
let npnvaf =            @npnv[vaf] 
let pnpbf  =            @pnpl[bf]
let pnpvaf =            @pnpl[vaf] 
let resq =              @rmodel[rsh]
echo                   "npnbf=$&npnbf npnvaf=$&npnvaf pnpbf=$&pnpbf pnpvaf=$&pnpvaf Ressq=$&resq" >> $outfile

foreach tempval     -25  27 125



set temp =  $tempval
foreach  vsval      1    2  
echo                    "~~~~~~~~~~~~~~~~~~~~~~~~~~~~~CONDITION_CHANGE~~~~~~~~~~~~~~~~~~~~~~~~~~~~~" >> $outfile

if (  $vsval =         1                                           )
op
*DC       SOURC1 VSTART VSTOP  VSTEP  SOURC2 START2 STOP2  STEP2
*======== ====== ====== ====== ====== ====== ====== ====== ======

dc        vcm    1.5    1.501  .001   vcc    5      5.001  .001
end                     ***<-- if $vsval

if ( $vsval =          2                                       )
dc        vcm    1.5    1.501  .001   vcc    10     10.001  .001
end                     ***<-- if $vsval 

let TT    =             @q1[temp]
let Vs    =             VCC[0]
let Vcmm  =             INP[0]
let Vout  =             OUT[0]
let Is    =             -vcc#branch[0]
let Iin   =             (@q1[ib] + @q2[ib])/2
let Ioff  =             abs(@q2[ib]-@q1[ib])
let vos   =             INP[0]-INN[0]
let avos  =             abs(INP[0]-INN[0])
let Cmrr  =             -db(abs(INP[0]-INN[0]-INP[1]+INN[1]))-60
let Psrr  =             -db(abs(INP[0]-INN[0]-INP[2]+INN[2]))-60
let Iout  =             (OUT[0]-VL[0])/10000
echo                    "Temp_C= $&TT  Vcc=$&Vs Vcm=$&Vcmm Vout=$&Vout Iout=$&Iout Is=$&Is"       >> $outfile
echo                    "vos  = $&vos  Iin =$&Iin  Ioff=$&Ioff Cmrr_dB=$&Cmrr Psrr_dB=$&Psrr "    >> $outfile
if (                     avos >1m or Iin >1u  or  Ioff > 1u or   cmrr < 90 or  psrr < 90   )             
echo                    "spec failure #########################################################"  >> $outfile
end                      ***<-- if avos >1m  etc.... 

*DC       SOURC1 VSTART VSTOP  VSTEP  SOURC2 START2 STOP2  STEP2
*======== ====== ====== ====== ====== ====== ====== ====== ======
dc        vin    1u     1001u  .001   vl     -29    31     30
let Avol  =             -db(abs(INP[2]-INN[2]-INP[3]+INN[3]))-60
let Ioutn =             (OUT[0]-VL[0])/10000
let Ioutp =             (OUT[4]-VL[4])/10000
echo                    "Avol_dB=$&Avol Ioutn = $&Ioutn  Ioutp = $&Ioutp"                         >> $outfile
if (                     Avol < 80  or Ioutn < 2m  or   Ioutp > -2m        )             
echo                    "spec failure #########################################################"  >> $outfile
end                      ***<-- if  Avol < 80u  etc....
  
*AC       DECADE NUMDEC FSTART FSTOP
*======== ====== ====== ====== ====== ====== ====== ====== ======
ac        dec    2      .1Meg  10Meg
let GBW  =               mag(out[0]/inn[0])/10
let Pmarg =              phase(inn[2]) -phase(out[2])
echo                    "GBW_MHz = $&GBW  Pmarg_deg = $&Pmarg   "                                 >> $outfile

end             **** <--foreach  vsval 
end             **** <--foreach  tempval
end             **** <--foreach  processval

echo                    "done"

.endc

*=======================MODELS=========================================================
*======================================================================================
.model        RMODEL       R (RSH=1000  TC1= 0.00138  TC2=0.0000000078 )

.model        npnv          npn ( 
+IS=1E-15     NF=1.0        BF=150      VAF=200        IKF=6e-03 
+CJE=1.7E-13  CJC=1.8E-13   CJS=2.9E-12 TF=1.6E-9                           
+XTB=2        EG=1.11       XTI=10      TNOM=25 

.model         pnpl         pnp ( 
+IS=7.91E-17   NF=1.01      BF=150      VAF=200       IKF=0.002 
+CJE=1.7E-13   CJC=1.8E-13  CJS=2.9E-12 TF=1.6E-9                                                   
+XTB=2         EG=1.11      XTI=7       TNOM=25 )                   

.end


